# Maintainer: Rexct <rexct1 at gmail dot com>
# Contributor: Ronald van Haren <ronald.archlinux.org>
# Contributor: Tom Gundersen <teg@jklm.no>
# Contributor: Mark Rosenstand <mark@archlinux.org>
pkgname=fuse
pkgver=2.9.3
pkgrel=1
pkgdesc="A library that makes it possible to implement a filesystem in a userspace program."
arch=('sh4')
url="http://fuse.sourceforge.net/"
license=('GPL2')
depends=('glibc')
makedepends=('pkg-config')
backup=(etc/fuse.conf)
source=(http://downloads.sourceforge.net/$pkgname/$pkgname-$pkgver.tar.gz
	'fuse.conf'
	)
options=(!libtool)
sha1sums=('94bd1974a9f2173ac3c2cf122f9fa3c35996b88e'
          '411a2f82aae6354cde551d2506aedf0fc1be85d8')

prepare() {
  cd "$pkgname-$pkgver"

  # fix building with glibc-2.14
  sed -i '1i#define _GNU_SOURCE' util/fusermount.c 
  
  sed -i "/MOUNT_FUSE_PATH=/s#/sbin#/usr/bin#" configure
  
  patch -Np1 -i "${srcdir}/fusexmp.c.patch"
}

build() {
  cd "$pkgname-$pkgver"
  ./configure --prefix=/usr --libdir=/usr/lib \
              --enable-lib --enable-util 
  make
}

package() {
  cd "$pkgname-$pkgver"

  make DESTDIR=${pkgdir} install

  # Remove init script in wrong path
  # Don't add our own for now, as fusectl fs oopses on 2.6.18
  # rm -rf ${pkgdir}/etc/init.d

  # install sample config file
  install -Dm644 ${srcdir}/fuse.conf ${pkgdir}/etc/fuse.conf

  # remove udev rules (is in the udev package}
  rm -rf ${pkgdir}/etc/udev

  # static device nodes are handled by udev
  rm -rf ${pkgdir}/dev
}
